Pensieve header: The tensorial double at $\epsilon^2=0$, Dror's edition.

The tensorial double at $\epsilon^2 = 0$

Utilities

(Alt) In[∘]:=
```
B2b = {B_i_^(p_.) → e^(-p b_i), B^(p_.) → e^(-p b)}; (* "B to lower b" *)
b2B = {e^(c_. b_i_ + d_.) :→ B_i^(-c) e^d, e^(c_. b + d_.) :→ B^(-c) e^d, e^ℰ_ :→ e^(Expand@ℰ)};
(* "b to upper B" *) CF[ℰ_] := ExpandDenominator@
  ExpandNumerator@Together[Expand[ℰ] //. e^x_ e^y_ :→ e^(x+y) /. e^x_ :→ e^(CF[x])];
Kδ /: Kδ_(i_,j_) := If[i === j, 1, 0];
```

(Alt) In[∘]:=
```
Unprotect[SeriesData];
SeriesData /: CF[sd_SeriesData] := MapAt[CF, sd, 3];
SeriesData /: Expand[sd_SeriesData] := MapAt[Expand, sd, 3];
SeriesData /: Simplify[sd_SeriesData] := MapAt[Simplify, sd, 3];
SeriesData /: Together[sd_SeriesData] := MapAt[Together, sd, 3];
SeriesData /: Collect[sd_SeriesData, specs__] := MapAt[Collect[#, specs] &, sd, 3];
Protect[SeriesData];
```

Zip and Bind

(Alt) In[∘]:=
```
𝔼 /: 𝔼[L1_, Q1_, P1_] ≡ 𝔼[L2_, Q2_, P2_] :=
  CF[L1 == L2] ∧ CF[Q1 == Q2] ∧ CF[Normal[P1 - P2] == 0];
𝔼 /: 𝔼[L1_, Q1_, P1_] 𝔼[L2_, Q2_, P2_] := 𝔼[L1 + L2, Q1 + Q2, P1 * P2];
```

(Alt) In[∘]:=
```
{b*, y*, a*, x*, z*} = {β, η, α, ξ, ζ};
{β*, η*, α*, ξ*, ζ*} = {b, y, a, x, z}; (u_i_)* := (u*)_i;
```

(Alt) In[∘]:=
```
Zip_{}[P_] := P; Zip_{ζ_,ζs___}[P_] := (Expand[P // Zip_{ζs}] /. f_. ζ^(d_.) :→ ∂_{ζ*,d} f) /. ζ* → 0
```

QZip implements the "Q-level zips" on $\mathbb{E}(L, Q, P) = Pe^{L+Q}$. Such zips regard the L variables as scalars.

(Alt) In[∘]:=
```
QZip_(ζs_List,simp_)@𝔼[L_, Q_, P_] := Module[{ζ, z, zs, c, ys, ηs, qt, zrule, Q1, Q2},
   zs = Table[ζ*, {ζ, ζs}];
   c = Q /. Alternatives @@ (ζs ⋃ zs) → 0;
   ys = Table[∂_ζ (Q /. Alternatives @@ zs → 0), {ζ, ζs}];
   ηs = Table[∂_z (Q /. Alternatives @@ ζs → 0), {z, zs}];
   qt = Inverse@Table[Kδ_(z,ζ*) - ∂_(z,ζ) Q, {ζ, ζs}, {z, zs}];
   zrule = Thread[zs → qt.(zs + ys)];
   Q2 = (Q1 = c + ηs.zs /. zrule) /. Alternatives @@ zs → 0;
   simp /@ 𝔼[L, Q2, Det[qt] e^(-Q2) Zip_ζs[e^Q1 (P /. zrule)]]];
QZip_(ζs_List) := QZip_(ζs,CF);
```

LZip implements the "L-level zips" on $\mathbb{E}(L, Q, P) = P e^{L+Q}$. Such zips regard all of $P e^{Q}$ as a single"P". Here the z's are b and α and the ζ's are β and a.

(Alt) In[∘]:=

```
LZip_{ζs_List,simp_}@𝔼[L_, Q_, P_] := Module[{ζ, z, zs, c, ys, ηs, lt, zrule, L1, L2, Q1, Q2},
   zs = Table[ζ*, {ζ, ζs}];
   c = L /. Alternatives @@ (ζs ⋃ zs) → 0;
   ys = Table[∂_ζ (L /. Alternatives @@ zs → 0), {ζ, ζs}];
   ηs = Table[∂_z (L /. Alternatives @@ ζs → 0), {z, zs}];
   lt = Inverse@Table[Kδ_{z,ζ*} - ∂_{z,ζ} L, {ζ, ζs}, {z, zs}];
   zrule = Thread[zs → lt.(zs + ys)];
   L2 = (L1 = c + ηs.zs /. zrule) /. Alternatives @@ zs → 0;
   Q2 = (Q1 = Q /. B2b /. zrule) /. Alternatives @@ zs → 0;
   simp /@ 𝔼[L2, Q2, Det[lt] e^{-L2-Q2} Zip_ζs[e^{L1+Q1} (P /. B2b /. zrule)]] //. b2B ];
LZip_{ζs_List} := LZip_{ζs,CF};
```

(Alt) In[∘]:=

```
Bind_{}[L_, R_] := L R;
Bind_{{is__}}[L_𝔼, R_𝔼] := Module[{n},
   Times[
      L /. Table[(v : B | b | a | x | y)_i → v_{n@i}, {i, {is}}],
      R /. Table[(v : β | α | ξ | η)_i → v_{n@i}, {i, {is}}]
    ] // LZip_{Flatten@Table[{β_{n@i},a_{n@i}},{i,{is}}]} // QZip_{Flatten@Table[{ξ_{n@i},y_{n@i}},{i,{is}}]} ];
B_{l_List}[L_, R_] := Bind_l[L, R]; B_{is___}[L_, R_] := Bind_{{is}}[L, R];
```

The two halves

(Alt) In[∘]:=

```
(*Hopf algebra on the a,x side*)
tam_{i_,j_->k_} := 𝔼[(α_i + α_j) a_k, (e^{-α_j} ξ_i + ξ_j) x_k, 1 + O[ϵ]^2]
taΔ_{i_->j_,k_} := 𝔼[α_i (a_j + a_k), ξ_i (x_j + x_k), 1 + ϵ ξ_i x_k (-a_j + 1/2 ξ_i x_j) + O[ϵ]^2]
taS_{i_} := 𝔼[-α_i a_i, - e^{α_i} ξ_i x_i, 1 - ϵ e^{α_i} ξ_i x_i (a_i + 1/2 e^{α_i} ξ_i x_i) + O[ϵ]^2]
taSi_{i_} := 𝔼[-α_i a_i, -e^{α_i} ξ_i x_i, 1 - ϵ e^{α_i} ξ_i x_i (a_i - 1 + 1/2 e^{α_i} ξ_i x_i) + O[ϵ]^2]
(*Hopf algebra on the y,b side*)
tbm_{i_,j_→k_} := 𝔼[(β_i + β_j) b_k, (η_i + η_j) y_k, 1 - ϵ η_j y_k β_i + O[ϵ]^2]
tbΔ_{i_->j_,k_} := 𝔼[β_i (b_j + b_k), η_i (e^{-b_k} y_j + y_k), 1 + 1/2 ϵ η_i^2 y_j y_k e^{-b_k} + O[ϵ]^2]
tbS_{i_} := 𝔼[-β_i b_i, -e^{b_i} η_i y_i, 1 - ϵ e^{b_i} η_i y_i (β_i + 1/2 e^{b_i} η_i y_i) + O[ϵ]^2]
tbSi_{i_} := 𝔼[-β_i b_i, -e^{b_i} η_i y_i, 1 - ϵ e^{b_i} η_i y_i (β_i - 1 + 1/2 e^{b_i} η_i y_i) + O[ϵ]^2]
```

First check that on the generators this agrees with our conventions in SLPortfolio.pdf with $\hbar = \gamma = 1$

(Alt) In[∘]:=
```
{
  "[a,x]" → ((𝔼[0, 0, a2 x1] ~B1,2~ tam1,2→1)⟦3⟧ - (𝔼[0, 0, a1 x2] ~B1,2~ tam1,2→1)⟦3⟧),
  "[b,y]" → ((𝔼[0, 0, y2 b1] ~B1,2~ tbm1,2→1)⟦3⟧ - (𝔼[0, 0, y1 b2] ~B1,2~ tbm1,2→1)⟦3⟧)
 } /. {z_1 → z} // Simplify
(Δ[#] → Simplify@Normal@Last[𝔼[0, 0, #1] ~B1~ taΔ1→1,2]) & /@ {a, x}
(Δ[#] → Simplify@Normal@Last[𝔼[0, 0, #1] ~B1~ tbΔ1→1,2]) & /@ {b, y}
{
  " S(a) = " ((𝔼[0, 0, a1] ~B1~ taS1)⟦3⟧),
  " S(x) = " ((𝔼[0, 0, x1] ~B1~ taS1)⟦3⟧),
  " S(b) = " ((𝔼[0, 0, b1] ~B1~ tbS1)⟦3⟧),
  " S(y) = " → ((𝔼[0, 0, y1] ~B1~ tbS1)⟦3⟧)
 } /. {z_1 → z} // Simplify
```

(Alt) Out[∘]= $\{[a,x] \to -x + O[\epsilon]^2,\ [b,y] \to -y\,\epsilon + O[\epsilon]^2\}$

(Alt) Out[∘]= $\{\Delta[a] \to a_1 + a_2,\ \Delta[x] \to x_1 + (1 - \epsilon\, a_1)\, x_2\}$

(Alt) Out[∘]= $\{\Delta[b] \to b_1 + b_2,\ \Delta[y] \to B_2\, y_1 + y_2\}$

(Alt) Out[∘]= $\{-\ S(a)\ =\ a + O[\epsilon]^2,\ -\ S(x)\ =\ x -\ S(x)\ =\ a\,x\,\epsilon + O[\epsilon]^2,$
$-\ S(b)\ =\ b + O[\epsilon]^2,\ S(y)\ =\ \to -\frac{y}{B} + O[\epsilon]^2\}$

Hopf algebra axioms on both sides separately

(co)-associativity on both sides

(Alt) In[∘]:=
```
{(taΔ1→1,2 ~B2~ taΔ2→2,3) ≡ (taΔ1→1,3 ~B1~ taΔ1→1,2),
 (tbΔ1→1,2 ~B2~ tbΔ2→2,3) ≡ (tbΔ1→1,3 ~B1~ tbΔ1→1,2),
 (tam1,2→1 ~B1~ tam1,3→1) ≡ (tam2,3→2 ~B2~ tam1,2->1),
 (tbm1,2→1 ~B1~ tbm1,3→1) ≡ (tbm2,3→2 ~B2~ tbm1,2->1)}
```

(Alt) Out[∘]= {True, True, True, True}

Δ is an algebra morphism

In[∘]:=
```
{tam1,2→1 ~B1~ taΔ1→1,2 ≡ (taΔ1→1,3 taΔ2→2,4) ~B1,2,3,4~ (tam3,4→2 tam1,2→1),
 tbm1,2→1 ~B1~ tbΔ1→1,2 ≡ (tbΔ1→1,3 tbΔ2→2,4) ~B1,2,3,4~ (tbm3,4→2 tbm1,2→1)}
```

Out[∘]= {True, True}

S is convolution inverse of id

In[∘]:=
```
{(taΔ1→1,2 ~B1~ taS1) ~B1,2~ tam1,2→1, (taΔ1→1,2 ~B2~ taS2) ~B1,2~ tam1,2→1}
{(tbΔ1→1,2 ~B1~ tbS1) ~B1,2~ tbm1,2→1, (tbΔ1→1,2 ~B2~ tbS2) ~B1,2~ tbm1,2→1}
```

Out[∘]= $\{\mathbb{E}[0, 0, 1 + O[\epsilon]^2],\ \mathbb{E}[0, 0, 1 + O[\epsilon]^2]\}$

Out[∘]= $\{\mathbb{E}[0, 0, 1 + O[\epsilon]^2],\ \mathbb{E}[0, 0, 1 + O[\epsilon]^2]\}$

Si is the inverse of S

In[∘]:= $\{\text{taSi}_1 \sim B_1 \sim \text{taS}_1 \equiv \mathbb{E}[a_1\,\alpha_1,\ x_1\,\xi_1,\ 1],\ \text{taS}_1 \sim B_1 \sim \text{taSi}_1 \equiv \mathbb{E}[a_1\,\alpha_1,\ x_1\,\xi_1,\ 1]\}$
$\{\text{tbSi}_1 \sim B_1 \sim \text{tbS}_1 \equiv \mathbb{E}[b_1\,\beta_1,\ y_1\,\eta_1,\ 1],\ \text{tbS}_1 \sim B_1 \sim \text{tbSi}_1 \equiv \mathbb{E}[b_1\,\beta_1,\ y_1\,\eta_1,\ 1]\}$

Out[∘]= {True, True}

Out[∘]= {True, True}

S is an algebra anti-(co)morphism

In[∘]:= $\{\text{tam}_{1,2\to1} \sim B_1 \sim \text{taS}_1 \equiv (\text{taS}_1\,\text{taS}_2) \sim B_{1,2} \sim \text{tam}_{2,1\to1},\ \text{tbm}_{1,2\to1} \sim B_1 \sim \text{tbS}_1 \equiv (\text{tbS}_1\,\text{tbS}_2) \sim B_{1,2} \sim \text{tbm}_{2,1\to1}\}$
$\{\text{taS}_1 \sim B_1 \sim \text{ta}\Delta_{1\to1,2} \equiv \text{ta}\Delta_{1\to2,1} \sim B_{1,2} \sim (\text{taS}_1\,\text{taS}_2),\ \text{tbS}_1 \sim B_1 \sim \text{tb}\Delta_{1\to1,2} \equiv \text{tb}\Delta_{1\to2,1} \sim B_{1,2} \sim (\text{tbS}_1\,\text{tbS}_2)\}$

Out[∘]= {True, True}

Out[∘]= {True, True}

Pairing

(Alt) In[∘]:= $\text{tP}_{i\_,j\_} := \mathbb{E}\left[\beta_i\,\alpha_j,\ \eta_i\,\xi_j,\ 1 + \frac{1}{4}\,\epsilon\,\eta_i^2\,\xi_j^2\right]$

In[∘]:= $\text{qfac}[k\_,\ q\_] := (1-q)^{-k}\ \text{QPochhammer}[q,\ q,\ k]\ //\ \text{FunctionExpand}$
$\text{qfe}[k\_] := \text{Normal}[\text{Series}[\text{qfac}[k,\ \text{E}\hat{}\rho],\ \{\rho,\ 0,\ 1\}]]\ /.\ \{\rho \to \epsilon\}$
$\text{Table}\left[\mathbb{E}\left[0,\ 0,\ y_1^r\,b_1^s\,a_2^t\,x_2^u\right] \sim B_{1,2} \sim \text{tP}_{1,2} \equiv \mathbb{E}[0,\ 0,\ \text{K}\delta_{r,u}\,\text{K}\delta_{s,t}\,\text{qfe}[r]\ s\,!],\right.$
$\left.\{r,\ 0,\ 4\},\ \{s,\ 0,\ 4\},\ \{t,\ 0,\ 4\},\ \{u,\ 0,\ 4\}\right]\ //\ \text{Flatten}\ //\ \text{Union}$

Out[∘]= {True}

Pairing axioms

In[∘]:= $\{(\text{tbm}_{1,2\to1}\ \mathbb{E}[\alpha_3\,a_3,\ \xi_3\,x_3,\ 1]) \sim B_{1,3} \sim \text{tP}_{1,3} \equiv$
$(\mathbb{E}[\beta_1\,b_1,\ \eta_1\,y_1,\ 1]\ \mathbb{E}[\beta_2\,b_2,\ \eta_2\,y_2,\ 1]\ \text{ta}\Delta_{3\to4,5}) \sim B_{1,4} \sim \text{tP}_{1,4} \sim B_{2,5} \sim \text{tP}_{2,5}$
$,\ (\text{tb}\Delta_{1\to1,2}\ \mathbb{E}[\alpha_3\,a_3,\ \xi_3\,x_3,\ 1]\ \mathbb{E}[\alpha_4\,a_4,\ \xi_4\,x_4,\ 1]) \sim B_{1,3} \sim \text{tP}_{1,3} \sim B_{2,4} \sim \text{tP}_{2,4} \equiv$
$(\mathbb{E}[\beta_1\,b_1,\ \eta_1\,y_1,\ 1]\ \text{tam}_{3,4\to3}) \sim B_{1,3} \sim \text{tP}_{1,3}\}$

Out[∘]= {True, True}

In[∘]:= $\{(\text{tbS}_1\ \mathbb{E}[\alpha_2\,a_2,\ \xi_2\,x_2,\ 1]) \sim B_{1,2} \sim \text{tP}_{1,2} \equiv (\mathbb{E}[\beta_1\,b_1,\ \eta_1\,y_1,\ 1]\ \text{taS}_2) \sim B_{1,2} \sim \text{tP}_{1,2},$
$(\text{tbSi}_1\ \mathbb{E}[\alpha_2\,a_2,\ \xi_2\,x_2,\ 1]) \sim B_{1,2} \sim \text{tP}_{1,2} \equiv (\mathbb{E}[\beta_1\,b_1,\ \eta_1\,y_1,\ 1]\ \text{taSi}_2) \sim B_{1,2} \sim \text{tP}_{1,2}\}$

Out[∘]= {True, True}

The Double

The double multiplication (should really bind the a’s and b’s separately)

(Alt) In[∘]:=
```
Block[{i, j, k}, tdm_{i_,j_→k_} =
  Simplify /@ Expand /@ ((𝔼[β_i b_i + α_j a_j, η_i y_i + ξ_j x_j, 1] (taΔ_{i→h1,h2} ~ B_{h2} ~ taΔ_{h2→h2,h3})
      (tbΔ_{j→t1,t2} ~ B_{t2} ~ tbΔ_{t2→t2,t3})) ~ B_{h3} ~ taSi_{h3} ~ B_{t1,h3} ~ (tP_{t1,h3}) ~
    B_{t3,h1} ~ (tP_{t3,h1}) ~ B_{h2,j,i,t2} ~ (tam_{h2,j→k} tbm_{i,t2→-k}) /. {u_-k :> u_k}) ]
```

(Alt) Out[∘]=
$$\mathbb{E}\Big[a_k(\alpha_i+\alpha_j)+b_k(\beta_i+\beta_j),\ y_k\left(\eta_i+e^{-\alpha_i}\eta_j\right)-\left(-1+B_k\right)\eta_j\xi_i+x_k\left(e^{-\alpha_j}\xi_i+\xi_j\right),$$
$$1+\frac{1}{4}e^{-\alpha_i-\alpha_j}\Big(-2y_k\eta_j\left(2e^{\alpha_j}\beta_i+\left(-2x_k+e^{\alpha_j}\left(-1+3B_k\right)\eta_j\right)\xi_i\right)-$$
$$e^{\alpha_i}\xi_i\left(x_k\left(4\beta_j+2\left(-1+3B_k\right)\eta_j\xi_i\right)-e^{\alpha_j}\eta_j\left(4a_kB_k+\left(1-4B_k+3B_k^2\right)\eta_j\xi_i\right)\right)\Big)\epsilon+O[\epsilon]^2\Big]$$

(Alt) In[∘]:=
```
(*Deriving tdS using tdm*)
Block[{i}, tdS_{i_} = ((tbSi_1 taS_2) ~ B_{1,2} ~ tdm_{2,1→i}) /. {z_1|2 → z_i}]
```

(Alt) Out[∘]=
$$\mathbb{E}\Big[-a_i\alpha_i-b_i\beta_i,\ \frac{-e^{\alpha_i}y_i\eta_i-e^{\alpha_i}B_ix_i\xi_i+e^{\alpha_i}\eta_i\xi_i-e^{\alpha_i}B_i\eta_i\xi_i}{B_i},$$
$$1+\frac{1}{4B_i^2}\Big(4e^{\alpha_i}B_iy_i\eta_i-4e^{\alpha_i}B_iy_i\beta_i\eta_i-2e^{2\alpha_i}y_i^2\eta_i^2-4e^{\alpha_i}a_iB_i^2x_i\xi_i-4e^{\alpha_i}B_i^2x_i\beta_i\xi_i-$$
$$4e^{\alpha_i}B_i\eta_i\xi_i+4e^{\alpha_i}a_iB_i\eta_i\xi_i+4e^{\alpha_i}B_i^2\eta_i\xi_i-4e^{2\alpha_i}B_ix_iy_i\eta_i\xi_i+4e^{\alpha_i}B_i\beta_i\eta_i\xi_i-$$
$$4e^{\alpha_i}B_i^2\beta_i\eta_i\xi_i+6e^{2\alpha_i}y_i\eta_i^2\xi_i-2e^{2\alpha_i}B_iy_i\eta_i^2\xi_i-2e^{2\alpha_i}B_i^2x_i^2\xi_i^2+6e^{2\alpha_i}B_ix_i\eta_i\xi_i^2-$$
$$2e^{2\alpha_i}B_i^2x_i\eta_i\xi_i^2-3e^{2\alpha_i}\eta_i^2\xi_i^2+4e^{2\alpha_i}B_i\eta_i^2\xi_i^2-e^{2\alpha_i}B_i^2\eta_i^2\xi_i^2\Big)\epsilon+O[\epsilon]^2\Big]$$

(Alt) In[∘]:=
```
(*Deriving tdΔ using tdm*)
Block[{i, j, k}, tdΔ_{i_→j_,k_} = (tbΔ_{i→3,1} taΔ_{i→2,4}) ~ B_{1,2,3,4} ~ (tdm_{3,4→k} tdm_{1,2→j}) ]
```

(Alt) Out[∘]=
$$\mathbb{E}\Big[a_j\alpha_i+a_k\alpha_i+b_j\beta_i+b_k\beta_i,\ y_j\eta_i+B_jy_k\eta_i+x_j\xi_i+x_k\xi_i,$$
$$1+\frac{1}{2}\left(B_jy_jy_k\eta_i^2-2a_jx_k\xi_i+x_jx_k\xi_i^2\right)\epsilon+O[\epsilon]^2\Big]$$

First check the double formulas on the generators agree with SL2Portfolio.pdf:

In[∘]:=
```
{
    "[a,y]" → ((𝔼[0, 0, y_2 a_1] ~ B_{1,2} ~ tdm_{1,2→1}) [[3]] - (𝔼[0, 0, y_1 a_2] ~ B_{1,2} ~ tdm_{1,2→1}) [[3]]),
    "[b,x]"  → ((𝔼[0, 0, x_2 b_1] ~ B_{1,2} ~ tdm_{1,2→1}) [[3]] - (𝔼[0, 0, x_1 b_2] ~ B_{1,2} ~ tdm_{1,2→1}) [[3]]),
    "xy-qyx" → ((𝔼[0, 0, x_1 y_2] ~ B_{1,2} ~ tdm_{1,2→1}) [[3]] - (1 + ϵ) (𝔼[0, 0, y_1 x_2] ~ B_{1,2} ~ tdm_{1,2→1}) [[3]])
  } /. {z_1 → z} // Expand // Factor
{
  "Δ(a)" → ((𝔼[0, 0, a_1] ~ B_1 ~ tdΔ_{1→1,2}) [[3]]),
  "Δ(x)" → ((𝔼[0, 0, x_1] ~ B_1 ~ tdΔ_{1→1,2}) [[3]]),
  "Δ(b)" → ((𝔼[0, 0, b_1] ~ B_1 ~ tdΔ_{1→1,2}) [[3]]),
  "Δ(y)" → ((𝔼[0, 0, y_1] ~ B_1 ~ tdΔ_{1→1,2}) [[3]])
} // Simplify
{
   "S(a)" → ((𝔼[0, 0, a_1] ~ B_1 ~ tdS_1) [[3]]),
   "S(x)" → ((𝔼[0, 0, x_1] ~ B_1 ~ tdS_1) [[3]]),
   "S(b)" → ((𝔼[0, 0, b_1] ~ B_1 ~ tdS_1) [[3]]),
   "S(y)" → ((𝔼[0, 0, y_1] ~ B_1 ~ tdS_1) [[3]])
} /. {z_1 → z} // Simplify
```

Out[∘]= $\{[a,y] \to -y + O[\epsilon]^2,\ [b,x] \to x\,\epsilon + O[\epsilon]^2,\ xy\text{-}qyx \to (1 - B) + a\,B\,\epsilon + O[\epsilon]^2\}$

Out[∘]= $\{\Delta(a) \to (a_1 + a_2) + O[\epsilon]^2,\ \Delta(x) \to (x_1 + x_2) - a_1\,x_2\,\epsilon + O[\epsilon]^2,\ \Delta(b) \to (b_1 + b_2) + O[\epsilon]^2,\ \Delta(y) \to (y_1 + B_1\,y_2) + O[\epsilon]^2\}$

Out[∘]= $\{S(a) \to -a + O[\epsilon]^2,\ S(x) \to -x - a\,x\,\epsilon + O[\epsilon]^2,\ S(b) \to -b + O[\epsilon]^2,\ S(y) \to -\frac{y}{B} + \frac{y\,\epsilon}{B} + O[\epsilon]^2\}$

Hopf algebra axioms on double

(co)-associativity

In[∘]:=
```
{(tdΔ_{1→1,2} ~ B_2 ~ tdΔ_{2→2,3}) ≡ (tdΔ_{1→1,3} ~ B_1 ~ tdΔ_{1→1,2}),
 (tdm_{1,2→1} ~ B_1 ~ tdm_{1,3→1}) ≡ (tdm_{2,3→2} ~ B_2 ~ tdm_{1,2->1})}
```

Out[∘]= {True, True}

Δ is an algebra morphism

In[∘]:=
```
tdm_{1,2→1} ~ B_1 ~ tdΔ_{1→1,2} ≡ (tdΔ_{1→1,3} tdΔ_{2→2,4}) ~ B_{1,2,3,4} ~ (tdm_{3,4→2} tdm_{1,2→1})
```

Out[∘]= True

S is convolution inverse of id

In[∘]:=
```
{(tdΔ_{1→1,2} ~ B_1 ~ tdS_1) ~ B_{1,2} ~ tdm_{1,2→1}, (tdΔ_{1→1,2} ~ B_2 ~ tdS_2) ~ B_{1,2} ~ tdm_{1,2→1}}
```

Out[∘]= $\{\mathbb{E}[0,\ 0,\ 1 + O[\epsilon]^2],\ \mathbb{E}[0,\ 0,\ 1 + O[\epsilon]^2]\}$

S is a (co)-algebra anti-morphism

In[∘]:=
```
{tdm_{1,2→1} ~ B_1 ~ tdS_1 ≡ (tdS_1 tdS_2) ~ B_{1,2} ~ tdm_{2,1→1},
 tdS_1 ~ B_1 ~ tdΔ_{1→1,2} ≡ tdΔ_{1→2,1} ~ B_{1,2} ~ (tdS_1 tdS_2)} // Expand
```

Out[∘]= {True, True}

R-matrix

(Alt) In[∘]:=

$$\mathbb{e}_{q\_,k\_}[x\_] := \mathbb{e}^\wedge\left(\sum_{j=1}^{k+1}\frac{(1-q)^j x^j}{j\,(1-q^j)}\right)$$

$$tR_{i\_,j\_} := \mathbb{E}\left[b_i\, a_j,\ y_i\, x_j,\ 1 - \epsilon\,\frac{1}{4}\,y_i^2\, x_j^2 + O[\epsilon]^2\right]$$

```
(*First two terms in Faddeev-Quesne formula*)
```

In[∘]:= $\text{Series}[\mathbb{e}_{q,1}[z]\ /.\ \{z\to y_i\, x_j,\ q\to 1+\rho\},\ \{\rho,0,1\}]\ /.\ \{\rho\to\epsilon\}$

Out[∘]= $\mathbb{e}^{x_j\, y_i} - \frac{1}{4}\left(\mathbb{e}^{x_j\, y_i}\, x_j^2\, y_i^2\right)\epsilon + O[\epsilon]^2$

Quasi-triangular axiom 1:

In[∘]:= $tR_{1,2} \sim B_1 \sim td\Delta_{1\text{->}1,3} \equiv (tR_{1,4}\, tR_{3,2}) \sim B_{2,4} \sim tdm_{2,4\to 2}$

Out[∘]= True

Quasi-triangular axiom 2:

In[∘]:= $((td\Delta_{1\text{->}1,2}\, tR_{3,4}) \sim B_{1,2,3,4} \sim (tdm_{1,3\to 1}\, tdm_{2,4\to 2})) \equiv$
$((td\Delta_{1\to 2,1}\, tR_{3,4}) \sim B_{1,2,3,4} \sim (tdm_{3,1\to 1}\, tdm_{4,2\to 2}))$

Out[∘]= True

Reidemeister 3:

In[∘]:= $((tR_{1,2}\, tR_{4,3}\, tR_{5,6}) \sim B_{1,4} \sim tdm_{1,4\to 1} \sim B_{2,5} \sim tdm_{2,5\to 2} \sim B_{3,6} \sim tdm_{3,6\to 3}) \equiv$
$((tR_{1,6}\, tR_{2,3}\, tR_{4,5}) \sim B_{1,4} \sim tdm_{1,4\to 1} \sim B_{2,5} \sim tdm_{2,5\to 2} \sim B_{3,6} \sim tdm_{3,6\to 3})$

Out[∘]= True

(Alt) In[∘]:= $\text{Block}\left[\{i, j\},\ \overline{tR}_{i\_,j\_} = \text{Expand}\ /@\ tR_{i,j} \sim B_j \sim tdS_j\right]$

(Alt) Out[∘]= $\mathbb{E}\left[-a_j\, b_i,\ -\frac{x_j\, y_i}{B_i},\ 1 + \left(-\frac{a_j\, x_j\, y_i}{B_i} - \frac{3\, x_j^2\, y_i^2}{4\, B_i^2}\right)\epsilon + O[\epsilon]^2\right]$

Reidemeister 2

In[∘]:= $\{(\overline{tR}_{1,2}\, tR_{3,4}) \sim B_{1,2,3,4} \sim (tdm_{1,3\to 1}\, tdm_{2,4\to 2}),\ (tR_{1,2}\, \overline{tR}_{3,4}) \sim B_{1,2,3,4} \sim (tdm_{1,3\to 1}\, tdm_{2,4\to 2})\}$

Out[∘]= $\{\mathbb{E}[0, 0, 1 + O[\epsilon]^2],\ \mathbb{E}[0, 0, 1 + O[\epsilon]^2]\}$

(Alt) In[∘]:=

```
dm = tdm; dS = tdS; dΔ = tdΔ; R = tR;
```

Deriving the Drinfeld element u and its inverse ui

(Alt) In[∘]:=
```
Block[{i}, {
  u_{i_} = R_{1,2} ~ B_1 ~ dS_1 ~ B_{1,2} ~ dm_{2,1→i},
  ui_{i_} := R_{1,2} ~ B_2 ~ dS_2 ~ B_2 ~ dS_2 ~ B_{1,2} ~ dm_{2,1→i}
 }]
```

(Alt) Out[∘]= $\left\{\mathbb{E}\left[-a_i\, b_i,\ -\frac{x_i\, y_i}{B_i},\ B_i + \frac{\left(-4\, a_i\, B_i^2 - 4\, B_i\, x_i\, y_i - 4\, a_i\, B_i\, x_i\, y_i - 3\, x_i^2\, y_i^2\right)\epsilon}{4\, B_i} + O[\epsilon]^2\right],\ \text{Null}\right\}$

u and ui are inverses

In[∘]:= $(u_1\ ui_2) \sim B_{1,2} \sim dm_{1,2\to 1}$

Out[∘]= $\mathbb{E}\left[0,\ 0,\ 1 + O[\epsilon]^2\right]$

The ribbon element v satisfies $v^2 = S(u)\, u$. The spinner C=uv^{-1}.
It is convenient to compute $z = S(u)\, u^{-1}$ which is something easy.

In[∘]:= $((u_1 \sim B_1 \sim dS_1)\ ui_2) \sim B_{1,2} \sim dm_{1,2\to 1}$

Out[∘]= $\mathbb{E}\left[0,\ 0,\ \frac{1}{B_1} + \frac{a_1\,\epsilon}{B_1} + O[\epsilon]^2\right]$

(* Needs fixing! *) So in our case S(u) = u z so S(u)u = $u^2\, z$ and v = u$z^{\frac{1}{2}}$ and finally C = uv^{-1} =
$z^{-\frac{1}{2}} = e^{t_1/2}(1 - \epsilon\, a_1)$.

(Alt) In[∘]:=
```
Block[{i},
 {CC_{i_} = 𝔼[0, 0, B_i^{1/2} e^{-ϵ a_i/2} + O[ϵ]^2],
  CC̄_{i_} = 𝔼[0, 0, B_i^{-1/2} e^{ϵ a_i/2} + O[ϵ]^2]
 }]
```

(Alt) Out[∘]= $\left\{\mathbb{E}\left[0,\ 0,\ \sqrt{B_i} - \frac{1}{2}\left(a_i\sqrt{B_i}\right)\epsilon + O[\epsilon]^2\right],\ \mathbb{E}\left[0,\ 0,\ \frac{1}{\sqrt{B_i}} + \frac{a_i\,\epsilon}{2\sqrt{B_i}} + O[\epsilon]^2\right]\right\}$

(Alt) In[∘]:=
```
Block[{i, j}, {
  Kink_{i_} = (R_{1,3} CC̄_2) ~ B_{1,2} ~ dm_{1,2→1} ~ B_{1,3} ~ dm_{1,3→i},
  Kink̄_{j_} = (R̄_{1,3} CC_2) ~ B_{1,2} ~ dm_{1,2→1} ~ B_{1,3} ~ dm_{1,3→j}
 }]
```

(Alt) Out[∘]= $\left\{\mathbb{E}\left[a_i\, b_i,\ x_i\, y_i,\ \frac{1}{\sqrt{B_i}} + \frac{\left(2\, a_i - x_i^2\, y_i^2\right)\epsilon}{4\sqrt{B_i}} + O[\epsilon]^2\right],\right.$
$\left.\mathbb{E}\left[-a_j\, b_j,\ -\frac{x_j\, y_j}{B_j},\ \sqrt{B_j} + \frac{\left(-2\, a_j\, B_j^2 - 4\, a_j\, B_j\, x_j\, y_j - 3\, x_j^2\, y_j^2\right)\epsilon}{4\, B_j^{3/2}} + O[\epsilon]^2\right]\right\}$

In[∘]:=
```
k2 = (R_{3,1} CC_2) ~ B_{1,2} ~ dm_{1,2→1} ~ B_{1,3} ~ dm_{1,3→i} /. e → E;
k4 = (R̄_{3,1} CC̄_2) ~ B_{1,2} ~ dm_{1,2→1} ~ B_{1,3} ~ dm_{1,3→j} /. e → E;
Simplify@{Kink_i ≡ k2, Kink̄_j ≡ k4, (Kink_i Kink̄_j) ~ B_{i,j} ~ dm_{i,j→1}}
```

Out[∘]= $\left\{\text{True},\ \text{True},\ \mathbb{E}\left[0,\ 0,\ 1 + O[\epsilon]^2\right]\right\}$

Reidemeister 2:

In[∘]:= $\left(R_{1,2}\,\overline{R}_{3,4}\right) \sim B_{1,3} \sim dm_{1,3\to1} \sim B_{2,4} \sim dm_{2,4\to2}$

Out[∘]= $\mathbb{E}\left[0,\ 0,\ 1 + O[\epsilon]^2\right]$

cyclic Reidemeister 2:

In[∘]:= $\left(R_{1,4}\,\overline{R}_{5,2}\,\overline{CC}_3\right) \sim B_{2,4} \sim dm_{2,4\to2} \sim B_{1,3} \sim dm_{1,3\to1} \sim B_{1,5} \sim dm_{1,5\to1} \equiv \overline{CC}_1$

Out[∘]= True

Trefoil

In[∘]:= Timing[
$Z = R_{1,5}\,R_{6,2}\,R_{3,7}\,\overline{CC}_4\,\overline{Kink}_8\,\overline{Kink}_9\,\overline{Kink}_{10}$;
Do[Print["doing ", r]; $Z = Z \sim B_{1,r} \sim dm_{1,r\to1}$, {r, 2, 10}];
Simplify@Z[[3]]]

doing 2

doing 3

doing 4

doing 5

doing 6

doing 7

doing 8

doing 9

doing 10

Out[∘]= $\left\{69.9531,\ \frac{B_1}{1 - B_1 + B_1^2} + \left(B_1\left(-B_1 + 2B_1^2 + 2B_1^4 + a_1\left(-1 + B_1 - B_1^3 + B_1^4\right) - 2x_1y_1 - B_1^3\left(3 + 2x_1y_1\right)\right)\epsilon\right) \Big/ \left(1 - B_1 + B_1^2\right)^3 + O[\epsilon]^2\right\}$